4. kapitola
Ako sa hýbu planéty

Spôsob, s pomocou ktorého sme v predošlej kapitole počítali pohyb hodeného vajca sa dá použiť aj v iných situáciach (ako naznačuje aj nadpis tejto kapitoly). Na to, aby sme ho mohli úspešne používať, sa ale musíme ešte nejaké veci naučiť.

Prvú vec, ktorú budeme potrebovať, je určovanie veľkosti vektora. O aute je totiž možné tvrdiť, že má rýchlosť $(45;40)\,km/h$, ale ak by sme chceli určiť jeho spotrebu po hodinovej jazde, alebo by sme chceli zistiť, či prekročilo maximálnu povolenú rýchlosť v obci, nepotrebujeme vedieť, aká je jeho rýchlosť v smere jednotlivých osí, ale aká je jeho rýchlosť smerom dopredu. Poznať veľkosť vektora sa môže hodiť aj vtedy, keď potrebujeme zistiť, ako sú nejaké dva body od seba ďaleko alebo vtedy, keď potrebujeme vektor, ktorý bude mať určený smer, ale veľkosť mu budeme chcieť upraviť.

Na zisťovanie dĺžky vektora sa používa stará dobrá Pytagorova veta. Tá hovorí, že keď nad stranami pravouhlého trojuholníka nakreslíme štvorce, tak obsah toho najväčšieho bude súčtom obsahov tých dvoch menších. Ak si označíme dve kratšie strany pri pravom uhle (odvesny) ako a a b a dlhšiu stranu oproti pravému uhlu (preponu) ako c, Pytagorova veta sa dá zapísať aj vo forme $a^2+b^2=c^2$. Z vektora a jeho dvoch súradníc však môžeme vyrobiť pravouhlý trojuholník kedykoľvek spôsobom, ktorý vidíte na obrázku 1 a to je presne to, čo na určenie dĺžky potrebujeme.

Na obrázku vidíte vektor $(5;2)$. Spolu s úsečkou dĺžky 5 rovnobežnou s osou x a s úsečkou dĺžky 2 rovnobežnou s osou y tvoria pravouhlý trojuholník. Ak budeme dĺžku vektora $\vec{v}$ označovať $|\vec{v}|$, potom podľa Pytagorovej vety musí platiť

$$|\vec{v}|^2=2^2+5^2=4+25=29$$

a teda

$$|\vec{v}|=\sqrt{29}\approx 5{,}3851648$$ [1]



Obrázok 1: Dĺžka vektora

Rovnako sa dá pre ľubovoľný vektor $\vec{u}=(x;y)$ ukázať, že jeho dĺžka bude $|\vec{u}|=\sqrt{x^2+y^2}$.

1 To $\approx$ sa číta „približne sa rovná“. Odmocnina z 29 má nekonečný desatinný rozvoj a keby sme ju chceli uviesť presne, vyplnila by zvyšok tejto knihy a ani to by nestačilo, takže uvádzame iba prvých sedem cifier za desatinnou čiarkou. Keby sme povedali „ten vektor je o kúsok kratší, ako päť a pol“, splnilo by to účel tiež.

Úloha 1: Zistite veľkosť vektorov $\vec{u}=(-4;3)$, $\vec{v}=(3;-1)$ a $\vec{w}=(-2;-2)$. Zapíšte to aj ako odmocniny (to je užitočné v situáciach, keď potrebujete počítať presne), aj vypočítajte, koľko to je, na kalkulačke (to je užitočné vtedy, keď potrebujete vedieť, koľko to je).

Úloha 2: Ak išlo auto rýchlosťou $(45;40)\,km/h$, išlo rýchlejšie, ako $50\,km/h$? Išlo rýchlejšie ako $60\,km/h$?

Predstavte si, že potrebujete nájsť vektor, ktorý má rovnaký smer, ako vektor $\vec{a}=(4;7)$, ale má dĺžku presne 10. Ako sa to dá spraviť? Vektor $\vec{a}$ má dĺžku $|\vec{a}|=\sqrt{4^2+7^2}=\sqrt{65}$. Tak ho touto dĺžkou vydeľme. Vektor

$$\frac{1}{\sqrt{65}}\vec{a}=\left(\frac{4}{\sqrt{65}};\frac{7}{\sqrt{65}}\right)$$

bude mať rovnaký smer a dĺžku 1. Keď tento vektor vynásobíme desiatimi, jeho smer sa stále nezmení, ale bude mať dĺžku 10. Hľadaný vektor je teda

$$\frac{10}{\sqrt{65}}\vec{a}=\left(\frac{40}{\sqrt{65}};\frac{70}{\sqrt{65}}\right)\approx(4{,}961389;8{,}682431)$$

Úloha 3: Nájdite taký násobok vektora $\vec{c}=(2;1)$, ktorý má dĺžku 5. Nájdite taký násobok vektora $\vec{d}=(1;1)$, ktorý má dĺžku 1. Výsledky znova napíšte aj v odmocninovom tvare, aj na kalkulačke vypočítajte, koľko to približne bude.

Ďalšia vec, ktorú budete potrebovať vedieť, je používanie tabuľkového kalkulátora.[2] Veľmi zjednodušene povedané, tabuľkový kalkulátor je program, ktorý má pracovnú plochu rozdelenú na jednotlivé bunky, do ktorých môžete zapisovať buď čísla, alebo vzorce, s pomocou ktorých sa čísla v bunkách vypočítajú. Na obrázku 2 môžete vidieť ukážku takého kalkulátora. Sú v ňom zatiaľ vyplnené len dve bunky. V bunke A1 je zapísaná hodnota 1 a v bunke C1 hodnota 2. Bunka C1 je momentálne aktívna, čo môžete vidieť podľa toho, že riadok 1 a stĺpec C sú farebne vyznačené, podľa toho, že je okolo bunky čierny rámik a nakoniec podľa toho, že vľavo hore nad oblasťou s bunkami je v roletovom menu napísané C1. Hodnotu, ktorá je v poli C1 môžete vidieť aj v širokom vstupnom riadku nad oblasťou s bunkami vpravo.



Obrázok 2: Tabuľkový kalkulátor

Ak do niektorej bunky chceme niečo vložiť, treba ju vybrať a začať písať. Ak do bunky chceme vložiť vzorec, treba na začiatku napísať znak = a potom vzorec, s pomocou ktorého sa má obsah bunky vypočítať. Na obrázku 3 sme do bunky A2 nastavili, aby zobrazovala súčin buniek A1 a C1. (Hviezdička znamená násobenie, znakmi $ sa zatiaľ nenechajte mýliť, o chvíľu sa ich význam vyjasní.)



Obrázok 3: Vkladanie vzorca

Keď po vložení vzorca stlačíte Enter, bunka zobrazí výsledok, ale vo vstupnom riadku sa stále bude zobrazovať spôsob výpočtu (je to vidno na obrázku 4). Ak by ste teraz zmenili obsah bunky A1 alebo C1, automaticky by sa zmenil aj obsah bunky A2.



Obrázok 4: Vložený vzorec

2 Ukážky v tejto knihe sú vytvorené s pomocou tabuľkového kalkulátora OpenOffice.org Calc. Celý kancelársky balík OpenOffice.org si môžete stiahnuť zadarmo a legálne z adresy http://sk.openoffice.org/

Teraz spravte malý pokus: Vyberte bunku A2, chyťte myšou ten malý čierny štvorček vpravo dole a potiahnite ho o štyri riadky nižšie. Výsledný efekt by mal byť rovnaký, ako môžete vidieť na obrázku 5.



Obrázok 5: Ťahanie

Čo sa vlastne udialo? Vzorec z bunky A2 sa zapísal aj do buniek pod ním. Lenže pritom sa zmenil. Ako môžete vidieť na obrázku 6, vzorec v bunke A3 nehovorí, aby sa vynásobili hodnoty A1 a C1, ale násobí hodnoty A2 a C1. Keď sme sa posunuli so vzorcom o políčko nižšie, o políčko nižšie sa posunul aj odkaz vo vzorci – namiesto A1 je v ňom A2.

Prečo sa ale neposunul aj odkaz na bunku C1? Prečo sa políčko A3 počíta ako A2 krát C1 a nie ako A2 krát C2? Práve kvôli tým znakom $. Tie kalkulátoru hovoria, že keď sa so vzorcom bude hýbať, údaje za znakom $ sa meniť nemajú. Pozor! Jeden dolár musí byť pred jednou súradnicou a jeden pred druhou.[3]



Obrázok 6: Zmenený vzorec

Takže v stĺpci A to momentálne funguje tak, že ďalšie políčko dostaneme z predošlého vynásobením bunkou C1, takže v ňom máme prvých pár mocnín dvojky. Ak hodnotu v C1 zmeníme, zmení sa aj obsah buniek v stĺpci A, ako môžete vidieť na obrázku 7.



Obrázok 7: Zmena hodnoty v C1

3 Vzhľadom na to, že sme vzorec ťahali iba dole, podstatný je len ten dolár pred číselnou súradnicou (teda ten pred jednotkou). Dolár pred súradnicou C by bol potrebný, keby sme chceli zabezpečiť, aby sa vzorec odkazoval stále na to isté políčko pri posune vpravo.

Po tomto rýchlokurze práce s tabuľkovým kalkulátorom sa môžeme pustiť do nebeskej mechaniky.

Na to, aby sme mohli počítať pohyby telies vo vesmíre, nám stačia dva relatívne jednoduché vzorce, s ktorými ste sa už mali možnosť stretnúť na fyzike. Prvý z nich

$$\vec{a}=\frac{\vec{F}}{m}$$

hovorí o tom, ako sa mení rýchlosť telesa, keď na neho pôsobí nejaká sila. Vo vzorci je $\vec{a}$ vektor, o ktorý sa rýchlosť telesa zmení, $\vec{F}$ je vektor sily, ktorá na teleso pôsobí a m je hmotnosť telesa (všimnite si, že toto nie je vektor, ale číslo, ktorým sa bude vektor sily deliť). Vzorec hovorí to, že čím väčšia sila, tým väčšia zmena rýchlosti a čím je teleso ťažšie, tým je náročnejšie jeho rýchlosť meniť. Tento vzorec je úplne skvelý. Stačí nám vedieť, aká sila na teleso pôsobí a akú má teleso hmotnosť a z toho už podobne ako v predošlej kapitole vieme vypočítať, akú rýchlosť získa a kde sa bude nachádzať. Mimochodom – tento skvelý vzorec je druhý Newtonov pohybový zákon.

Druhý vzorec sa tiež volá podľa Newtona – je to Newtonov gravitačný zákon. Hovorí o tom, akou silou na seba dve telesá vo vesmíre pôsobia:

$$F=\kappa\frac{m_1 . m_2}{r^2}$$

V tomto vzorci sú m_1 a m_2 hmotnosti telies, r je ich vzdialenosť a κ je nejaká konštanta – konkrétne 0,000 000 000 066 74. Fyzici to radšej zapisujú v tvare $6{,}674\times10^{-11}$, aby nemuseli písať toľko núl.[4] Do tabuľkového kalkulátora (a do niektorých kalkulačiek) sa takéto číslo zapisuje ako 6,674E–11. Tento vzorec hovorí, že čím sú telesá ťažšie, tým je sila väčšia, ale že tá sila klesá s druhou mocninou vzdialenosti telies. (To je dané tým, že žijeme v trojrozmernom priestore. Keby sme žili v štvorrozmernom, klesala by s treťou mocninou.)

Všimnite si, že nad písmenkom F v tomto prípade chýba šípka. Newtonov gravitačný zákon nám totiž prezradí, aká bude sila veľká, ale nepovie nám, ktorým smerom bude pôsobiť. Vektor sily ale bude smerovať od jedného telesa k druhému[5] a keďže tým pádom poznáme jeho smer aj veľkosť, nič nám nebráni vypočítať si ho spôsobom uvedeným na začiatku tejto kapitoly.

Poďme sa teraz venovať konkrétnemu prípadu – Mesiacu a jeho pohybu okolo Zeme. Kvôli jednoduchosti budeme predpokladať, že Zem stojí na mieste.[6] Na začiatku poznáme rýchlosť $\vec{v}$ Mesiaca a jeho polohu. Potrebujeme zistiť, kam sa Mesiac presunie, kým uplynie čas Δt.

4 Z toho zápisu rovno vedia, že tá prvá šestka bude na jedenástom mieste za desatinnou čiarkou.

5 Vnímavejší čitateľ si možno všimol, že nie je jasné, od ktorého telesa ku ktorému ten vektor vlastne ide. Odpoveď sa skrýva v treťom Newtonovom pohybovom zákone, ktorý hovorí, že žiadna sila sa nevyskytuje samostatne a že keď jedno teleso pôsobí na druhé nejakou silou, tak to druhé pôsobí na prvé presne opačnou silou. Takže aj v tomto prípade pôjde jeden vektor od prvého telesa k druhému a druhý vektor od druhého telesa k prvému.

6 Čo nie je celkom pravda – rovnako, ako Zem vplýva na dráhu Mesiaca, vplýva Mesiac na dráhu Zeme. Výpočet si môžete neskôr upraviť tak, aby počítal aj so zmenou polohy Zeme.



Obrázok 8: Pohyb Mesiaca okolo Zeme

Najprv poďme zistiť, ako sa za určený čas zmení rýchlosť. Keby na Mesiac nepôsobila žiadna sila, Mesiac by mal stále rýchlosť $\vec{v}$ a skrátka by nám uletel (na obrázku 8 by zmizol niekde v smere patričnej šípky). Našťastie na neho pôsobí gravitačná sila $\vec{F}$, ktorá jeho rýchlosť zmení každú sekundu o $\vec{F}/m_M$, kde to m_M je hmotnosť Mesiaca. Za čas Δt sa teda rýchlosť zmení o vektor $(\vec{F}/m_M).\Delta t$. Nová rýchlosť teda bude

$$\vec{v}_n = \vec{v} + \frac{\vec{F}}{m_M}.\Delta t$$

Táto rýchlosť je na obrázku 8 naznačená prerušovanou čiarou. Za čas Δt sa Mesiac posunie touto novou rýchlosťou o $\vec{v}_n.\Delta t$. Poznáme teda novú polohu Mesiaca aj jeho novú rýchlosť. A úplne rovnakým spôsobom môžeme znovu vyrátať ďalšiu novú rýchlosť a potom novú polohu Mesiaca a potom znovu a znovu atď.

Rátať takéto niečo ručne alebo čo i len na kalkulačke je samozrejme práca pre vraha. Našťastie máme tabuľkový kalkulátor. Poďme teda kalkulovať. Predtým si ale na internete[7] zistíme nejaké údaje, ktoré potrebujeme. V prvom rade potrebujeme vedieť hmotnosť Zeme $m_Z = 5{,}9736 \times 10^{24}\,kg$ a hmotnosť Mesiaca $m_M = 7{,}3477 \times 10^{22}\,kg$ (všimnite si, že Zem je výrazne ťažšia). Ako počiatočnú rýchlosť použijeme vektor $(0, v)$, kde $v = 1022\,m/s$ je priemerná obežná rýchlosť Mesiaca okolo Zeme, Zem dáme do počiatku súradnicovej sústavy a ako počiatočnú polohu Mesiaca použijeme bod $[r; 0]$, kde $r = 384\,399\,000\,m$ je priemerná vzdialenosť Mesiaca od Zeme.

E7 | $f(x)$ Σ = | 0

	A	B	C	D	E
1	Kappa:	6,67E-011			
2	m Zeme:	5,97E+024			
3	m Mesiaca:	7,35E+022			
4	Interval	86400			
5					
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y
7	0	0	1022	384399000	0
8					

Obrázok 9: Úvodné údaje

Všetky potrebné čísla si zapíšeme do tabuľky, ako to môžete vidieť na obrázku 9. Na políčku B1 budeme mať gravitačnú konštantu. (Zápis `6,67E-11` znamená $6{,}67 \times 10^{-11}$. Ak tam napíšete `6,674E-11`, tabuľkový kalkulátor zobrazí za desatinnou čiarkou iba dve miesta, ako je vidieť na obrázku, ale pamätať si to bude presne. Ak chcete vidieť viac desatinných miest, dá sa to bunke nastaviť.) Na políčku B2 bude hmotnosť Zeme a na políčku B3 hmotnosť Mesiaca. Na políčku B4

7 Odporúčaná adresa je http://en.wikipedia.org

bude nastavený časový interval, v akom budeme počítať novú polohu Mesiaca oproti starej. Momentálne je nastavený na 86400 sekúnd, čo je presne jeden deň.

Uvedené štyri čísla sa počas celého výpočtu meniť nebudú. Preto vždy, keď budeme niektoré z nich potrebovať použiť, dáme pred súradnice jeho bunky znaky $.

Okrem toho si ostatné známe údaje pripravíme do jednotlivých stĺpcov. V stĺpci A budeme mať uloženú informáciu o aktuálnom čase. Údaj budeme zobrazovať v dňoch, nech sa v tom dá lepšie orientovať. Stĺpce B a C budú určovať vektor rýchlosti Mesiaca a v stĺpcoch D a E budú súradnice polohy Mesiaca. Skontrolujte si, či vaša práca vyzerá podobne, ako na obrázku 9.

Prvá vec, ktorú bude treba vypočítať, je aktuálna vzdialenosť Mesiaca od Zeme. Vyhradíme pre ňu stĺpec F (pozrite obrázok 10). Vypočítame ju ako dĺžku vektora, ktorý vedie od Zeme k Mesiacu, teda ako $\sqrt{(x_M-0)^2+(y_M-0)^2}$, kde $[x_M; y_M]$ sú súradnice Mesiaca. (Ak by Zem nebola v počiatku súradnicovej sústavy, tie nuly by bolo treba nahradiť jej súradnicami.)

V tabuľkovom kalkulátore sa $\sqrt{x_M^2+y_M^2}$ zapíše ako `=SQRT(D7^2+E7^2)` (všimnite si vstupný riadok na obrázku 10). `SQRT` je funkcia pre výpočet druhej odmocniny, znak `^2` znamená umocňovanie na druhú a v bunkách D7 a E7 sú uložené súradnice Mesiaca. Po stlačení klávesy Enter by situácia mohla vyzerať rovnako, ako na obrázku 10.


F7 =SQRT(D7^2+E7^2)

	A	B	C	D	E	F
1	Kappa:	6,67E-011				
2	m Zeme:	5,97E+024				
3	m Mesiaca:	7,35E+022				
4	Interval	86400				
5						
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť
7	0	0	1022	384399000	0	384399000

Obrázok 10: Vzdialenosť Mesiaca od Zeme

Ďalšia vec, ktorú bude treba vypočítať, je veľkosť gravitačnej sily. Tomuto údaju bude rezervovaný stĺpec G (obrázok 11). Počítať budeme podľa Newtonovho gravitačného zákona, to znamená, že vynásobíme hmotnosti Mesiaca a Zeme a gravitačnú konštantu a výsledok vydelíme druhou mocninou vzdialenosti. Do G7 teda napíšeme `= $B$1 * $B$2 * $B$3 / F7^2`.

Všimnite si, že pri odkaze na bunku F7, v ktorej je uložená aktuálna vzdialenosť Mesiaca od Zeme sme nepoužili znaky $. Ak totiž budeme potrebovať vypočítať o riadok nižšie gravitačnú silu, ktorou na seba budú obe telesá pôsobiť o 24 hodín, vzdialenosť medzi nimi už bude iná a budeme teda potrebovať údaj z bunky F8 a nie z bunky F7. Poloha ostatných buniek v použitom vzorci však musí zostať zachovaná.


G7 = B1 * B2 * B3 / F7^2

	A	B	C	D	E	F	G
1	Kappa:	6,67E-011					
2	m Zeme:	5,97E+024					
3	m Mesiaca:	7,35E+022					
4	Interval	86400					
5							
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť	Sila
7	0	0	1022	384399000	0	384399000	1,98E+020

Obrázok 11: Gravitačná sila

Keď sme zistili veľkosť sily (mimochodom – všimnite si, aké obrovské číslo to je), potrebujeme ešte vyrátať jej vektor. Ten bude smerovať od Mesiaca k Zemi, bude teda násobkom vektora $(0-x_M;0-y_M)=(-x_M;-y_M)$. Treba iba zariadiť, aby mal takú veľkosť, akú sme vypočítali v stĺpci G. Na to ale stačí vydeliť ho jeho veľkosťou (tú už máme vypočítanú v stĺpci F) a vynásobiť údajom zo stĺpca G.

Vektor sily uložíme v stĺpcoch H a I. V bunke H7 bude vzorec = `(-D7 / F7) * G7` a v bunke I7 bude vzorec = `(-E7 / F7) * G7`. Výsledok môžete vidieť na obrázku 12.

I7 | = (-E7 / F7) * G7

	A	B	C	D	E	F	G	H	I
1	Kappa:	6,67E-011							
2	m Zeme:	5,97E+024							
3	m Mesiaca:	7,35E+022							
4	Interval	86400							
5									
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť	Sila	Sila X	Sila Y
7	0	0	1022	384399000	0	384399000	1,98E+020	-1,98E+020	0

Obrázok 12: Vektor sily

Všetky potrebné informácie o sústave Zem – Mesiac v čase 0 sme teda vypočítali a prvý riadok máme úplný. Prejdime na druhý riadok. V ňom budeme musieť vypočítať novú rýchlosť a novú polohu z údajov, ktoré sú o riadok vyššie. Stačí nám vyplniť teda bunky A8 až E8. Recept na počítanie ostatných buniek je už uložený v siedmom riadku. Nebudeme ich musieť znovu počítať, bude stačiť, keď bunky F7 až I7 vyberieme a potiahneme nadol čierny štvorček v pravom dolnom rohu výberu. Poďme ale najprv vyplniť zostávajúce bunky.

Hodnotu v bunke A8 vypočítame tak, že k bunke A7 pripočítame údaj z B4 vydelený 86400, pretože v B4 sa nachádza interval v sekundách, my chceme mať v stĺpci A časový údaj v dňoch a deň má 86400 sekúnd. Patričný vzorec môžete vidieť vo vstupnom riadku na obrázku 13. V bunke A8 sa zobrazí hodnota 1. Keby sme ale kvôli spresneniu výpočtu zmenili interval na 43200 sekúnd (čo je pol dňa), bola by tam hodnota 0,5.

A8 | =A7 + (B4 / 86400)

	A	B	C	D	E	F	G	H	I
1	Kappa:	6,67E-011							
2	m Zeme:	5,97E+024							
3	m Mesiaca:	7,35E+022							
4	Interval	86400							
5									
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť	Sila	Sila X	Sila Y
7	0	0	1022	384399000	0	384399000	1,98E+020	-1,98E+020	0
8	1								

Obrázok 13: Čas v dňoch

Teraz potrebujeme vypočítať novú rýchlosť. Vektor rýchlosti sa každú sekundu zmení o $\vec{F}/m_M$. Prvá súradnica rýchlosti sa teda za čas Δt uvedený v bunke B4 zmení o $(x_F/m_M).\Delta t$. Do bunky B8 treba teda napísať = `B7 + (H7 / $B$3) * $B$4` (obrázok 14). Rovnako je to s druhou súradnicou. V bunke C8 bude teda = `C7 + (I7 / $B$3) * $B$4`.

B8 | = B7 + (H7 / B3) * B4

	A	B	C	D	E	F	G	H	I
1	Kappa:	6,67E-011							
2	m Zeme:	5,97E+024							
3	m Mesiaca:	7,35E+022							
4	Interval	86400							
5									
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť	Sila	Sila X	Sila Y
7	0	0	1022	384399000	0	384399000	1,98E+020	-1,98E+020	0
8	1	-233,12	1022						

Obrázok 14: Nová rýchlosť

Keď poznáme novú rýchlosť, stačí už len vypočítať novú polohu. Bude to predošlá poloha plus nová rýchlosť krát čas. Do bunky D8 teda napíšeme vzorec `= D7 + (B8 * $B$4)` (obrázok 15) a v bunke E8 bude vzorec `= E7 + (C8 * $B$4)`.



D8 | = D7 + (B8 * B4)

	A	B	C	D	E	F	G	H	I
1	Kappa:	6,67E-011							
2	m Zeme:	5,97E+024							
3	m Mesiaca:	7,35E+022							
4	Interval	86400							
5									
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť	Sila	Sila X	Sila Y
7	0	0	1022	384399000	0	384399000	1,98E+020	-1,98E+020	0
8	1	-233,12	1022	364257800,69	88300800				

Obrázok 15: Nová poloha

Bunky v stĺpcoch F až I (vzdialenosť Mesiaca od Zeme a sila, ktorou Zem na Mesiac pôsobí) sa počítajú z buniek A až E toho istého riadku. Spôsob ich výpočtu sme už zadali v siedmom riadku. Teraz stačí vybrať bunky F7 až I7 (napríklad tak, že kliknete do bunky F7 a kliknete so stlačenou klávesou Shift do bunky I7), chytiť myšou malý čierny štvorček vpravo dole a potiahnuť o riadok nižšie (obrázok 16). Patrične upravené vzorce sa prepíšu aj do ôsmeho riadku.



F7:I8 | = (-E7 / F7) * G7

	A	B	C	D	E	F	G	H	I
1	Kappa:	6,67E-011							
2	m Zeme:	5,97E+024							
3	m Mesiaca:	7,35E+022							
4	Interval	86400							
5									
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť	Sila	Sila X	Sila Y
7	0	0	1022	384399000	0	384399000	1,98E+020	-1,98E+020	0
8	1	-233,12	1022	364257800,69	88300800	374807652,86	2,09E+020	-2,03E+020	-4,91E+019

Obrázok 16: Opakovanie výpočtu pre novú polohu Mesiaca

Skvelé. V ôsmom riadku máme teraz každú bunku vypočítanú s pomocou buniek z predošlého riadku alebo s pomocou buniek z toho istého riadku. Ak podobne ako v predošlom kroku aktivujeme celý ôsmy riadok – teda bunky A8 až I8 a budeme ťahať nadol, tabuľkový kalkulátor nám bude počítať údaje (z ktorých nás zvlášť zaujíma poloha v stĺpcoch D a E) pre jednotlivé časy. Nebojte sa potiahnuť čierny štvorček nadol o niekoľko desiatok riadkov.



A8:I10 | = (-E8 / F8) * G8

	A	B	C	D	E	F	G	H	I
1	Kappa:	6,67E-011							
2	m Zeme:	5,97E+024							
3	m Mesiaca:	7,35E+022							
4	Interval	86400							
5									
6	Čas	Rýchlosť X	Rýchlosť Y	Poloha X	Poloha Y	Vzdialenosť	Sila	Sila X	Sila Y
7	0	0	1022	384399000	0	384399000	1,98E+020	-1,98E+020	0
8	1	-233,12	1022	364257800,69	88300800	374807652,86	2,09E+020	-2,03E+020	-4,91E+019
9	2	-471,41	964,23	323527692,22	171610582,25	366224466,1	2,18E+020	-1,93E+020	-1,02E+020
10	3	-698,3	843,89	263194733,52	244522316,28	359252878,78	2,27E+020	-1,66E+020	-1,54E+020

Obrázok 17: Pohyb Mesiaca

Tabuľkové kalkulátory majú ešte jednu výhodu – z vypočítaných hodnôt vedia robiť grafy. Na to stačí vybrať hodnoty, ktoré chcete pre graf použiť (v našom prípade bunky D7 až E36) a stlačiť ikonu [8]. Keď sa potom preklikáte nejakými dialógovými oknami (ako typ grafu v tomto prípade treba zvoliť XY), dostanete graf podobný tomu, ktorý môžete vidieť na obrázku 18.



Obrázok 18: Pohyb Mesiaca okolo Zeme

A na záver úlohy:

Úloha 4: Koľko dní trvalo Mesiacu vo vašom výpočte, kým obehol okolo Zeme?

Úloha 5: Akú najmenšiu a akú najväčšiu vzdialenosť mal Mesiac od Zeme? Porovnajte tieto údaje s nameranými údajmi, ktoré nájdete na internete. Akú chybu vykazuje naša metóda?

Úloha 6: Ako sa bude metóda správať, keď budete polohu počítať raz za dvanásť hodín? (Stačí patrične zmeniť bunku B4.) Ako vyjde minimálna a maximálna vzdialenosť a obežná doba? Aká bude chyba? Ako to bude vyzerať, keď budete novú polohu počítať raz za hodinu?

Úloha 7: Ako by vyzerala dráha Mesiaca, keby počiatočná rýchlosť nebola 1022 m/s, ale 800 m/s? A keby bola 500 m/s? Alebo 1200 m/s? Alebo 1500 m/s? Vyskúšajte meniť hodnoty v bunke C7 a pozorujte na grafe, čo to robí s dráhou Mesiaca.

Úloha 8: Ako by vyzerala dráha Mesiaca, keby jeho počiatočná vzdialenosť bola jeden a polkrát väčšia? A keby bola dvakrát väčšia?

Úloha 9: Ako by vyzerala dráha Mesiaca, keby bol dvakrát ťažší? Keby bol desaťkrát ťažší?

Úloha 10: Použite vytvorenú tabuľku na výpočet pohybu Marsu okolo Slnka. Hmotnosť Slnka je $1{,}9891\times10^{30}\,kg$, hmotnosť Marsu je $6{,}4185\times10^{23}\,kg$, stredná vzdialenosť od Slnka je $227\,939\,100\,000\,m$ a priemerná rýchlosť obehu je $24\,077\,m/s$.

8 OpenOffice.org má viaceré schémy ikon, takže je možné, že u vás bude ikona vyzerať trochu inak.

Všetky predošlé úlohy sa dajú riešiť tak, že v tabuľke, ktorú sme vytvorili, zmeníme jednu až štyri bunky. Nasledujúce úlohy sú náročnejšie. Každá z nich stojí na vzťahu $\vec{a}=\vec{F}/m$, ktorý sme použili aj na sústavu Zem – Mesiac. Treba ale prerobiť spôsob výpočtu a zistiť si potrebné údaje. Preto úlohy nazveme honosne „projekty“. Riešte ich v skupinách po troch až štyroch ľuďoch.

Projekt 1: Upravte výpočet, aby sa brala do úvahy aj sila, ktorou pôsobí Mesiac na Zem a aby sa tým pádom nemenila len poloha Mesiaca, ale aj poloha Zeme. Sú výsledky presnejšie v porovnaní s predošlým prístupom, alebo sme iba do výpočtu pridali ďalšie chyby? (Porovnajte napr. minimálnu a maximálnu vzdialenosť s reálne nameranými hodnotami.) Skúste v tomto prípade vyriešiť úlohy 7, 8 a 9 z tejto kapitoly.

Projekt 2: Nech je poloha Zeme na súradniciach $[0;0]$ a poloha Mesiaca $[384\,399\,000;0]$. Veliteľský a lunárny modul misie Apollo 11 – dohromady vážia 46769 kg – vyrážajú z obežnej dráhy okolo Zeme na cestu k Mesiacu. Vzdialenosť od povrchu je 100,2 námornej míle, teda v prepočte na metre a pripočítaní polomeru Zeme začína misia cestu z bodu [6564106;0]. Akou rýchlosťou a ktorým smerom treba Apollo 11 vypustiť, aby sa po prílete k Mesiacu ustálilo na obežnej dráhe s polomerom asi 1 900 kilometrov (čo je asi 88 námorných míľ nad povrchom Mesiaca)? Cestou na ňu pôsobí aj gravitácia Zeme aj gravitácia Mesiaca, takže započítajte pri zmene rýchlosti obidve tieto sily. V tabuľkovom kalkulátore vytvorte tabuľku, ktorá vám to bude počítať a experimentujte s rôznymi počiatočnými smermi a rýchlosťami. Aký najlepší výsledok sa vám podarí dosiahnuť? (Ak dosiahnete obežnú dráhu pod povrchom Mesiaca, **nie je** to dobrý výsledok. Ak vám sonda uletí do voľného vesmíru, tiež to **nie je** dobrý výsledok. Chýbajúce údaje si zistite, ako viete.)

Projekt 3: Vypočítajte dráhy štyroch najväčších Jupiterovych mesiacov okolo Jupitera. Zistite, ako dosiahnuť, aby sa všetky zobrazili v jednom grafe. Skúste zvážiť, ako by sa na každé teleso dal vždy vypočítať vplyv ostatných štyroch telies.

Projekt 4: Vo výpočtoch v predošlej kapitole sme zanedbávali odpor vzduchu. Sila, ktorou pôsobí vzduch na letiacu guľu je $\vec{F}=-0{,}195|\vec{v}|^2 r^2 \hat{v}$, kde $|\vec{v}|$ je veľkosť rýchlosti, r je polomer gule a $\hat{v}$ je vektor, ktorý má rovnaký smer ako $\vec{v}$, ale dĺžku má 1.[9] To, akú zmenu rýchlosti spôsobí táto sila, samozrejme závisí od hmotnosti telesa. Zistite, akú dráhu opíše tenisová loptička ($r=3{,}35\,cm$, $m=56\,g$) ak ju vystrelíte z kanóna rýchlosťou $v=300\,m/s$ pod uhlom 45°. Akú dráhu opíše rovnako veľká železná guľa? (Hustota železa je $7874\,kg/m^3$.) Porovnajte obe dráhy so situáciou, keď sme odpor vzduchu zanedbali.

9 Ak chcete vedieť, ako to funguje pri iných telesách, než guliach, pozrite si článok http://en.wikipedia.org/wiki/Drag_(physics)